
Subject: Re: TOTAL gives totally different result on identical array

Posted by [Foldy Lajos](#) on Fri, 08 Jul 2011 15:44:39 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Fri, 8 Jul 2011, Liam Gumley wrote:

```
> function kahansum, input
> sum = 0.0
> c = 0.0
> for i = 0L, n_elements(input) - 1 do begin
>   y = input[i] - c
>   t = sum + y
>   c = (t - sum) - y
>   sum = t
> endfor
> return, sum
> end
```

It will be very slow. But it's IDL, vectorize it!

```
; test.pro begin
```

```
function kahan_sum, input
sum=0.0
c=0.0
for i=0I, n_elements(input)-1 do begin
    y=input[i]-c
    t=sum+y
    c=(t-sum)-y
    sum=t
endfor
return, sum
end
```

```
function kahan_sum_1000, input
sum=fltarr(1000)
c=fltarr(1000)
for i=0I, n_elements(input)-1,1000 do begin
    y=input[i:i+999]-c
    t=sum+y
    c=(t-sum)-y
    sum=t
endfor
return, kahan_sum(sum)
end
```

```
pro test
vec=findgen(100000)

t1=systime(1)
s1=kahan_sum(vec)
t1=systime(1)-t1
print, 't1:', t1

t2=systime(1)
s2=kahan_sum_1000(vec)
t2=systime(1)-t2
print, 't2:', t2
print, 'ratio:', t1/t2
print, 'sums and diff:', s1, s2, s2-s1

end

; test.pro end

IDL> test
t1:  0.056817055
t2:  0.0012319088
ratio:  46.121153
sums and diff: 4.99995e+09 4.99995e+09  0.00000
IDL>
```

(The general case is left to the reader.)

regards,
Lajos
